
ABSTRACT 
In this paper, we demonstrate that many hypothesis testing problems in clinical pharmacology studies can be 
reformulated as a one- or two-sample t-test or equivalence test. Thereby one can use SAS® Proc Power to perform 
the power analysis.  Sample programs for different hypothesis testing problems are presented in a variety of study 
design settings. Compared to other commercial software such as NQuery and PASS, using Proc Power offers several 
advantages. First, one often obtains variance estimates by analyzing relevant clinical data in SAS, thus it is more 
streamlined and convenient to conduct power analysis in SAS as well. Second, the documentation in SAS is more 
comprehensive than those in NQuery and PASS; this helps the users better understand the procedure and hence the 
users are less likely to make mistakes. Third, as demonstrated in the examples, one can take advantage of the 
programming capabilities in SAS and make inputting parameters for power analysis more straightforward. Last, but 
not least, using SAS can better help the users keep a record of the power analysis, which is helpful for adapting the 
code for similar problems in the future. 

1. INTRODUCTION 
The main objective of pharmacology studies is often to characterize the pharmacokinetics (PK) profile of the 
candidate drug, which is represented by the concentration-time curve and parameters derived from it, such as AUC, 
Cmax, Ctrough, Tmax, and apparent t1/2. There are a variety of clinical pharmacology studies, including first-in-man, 
multiple dose, single dose elderly, drug-drug interaction, bioequivalence, QTc, hepatic insufficiency, etc. Study 
designs such as parallel group, fixed sequence, and crossover have all been applied in those studies. For example, 
multiple dose studies and hepatic insufficiency studies often use a parallel group design; drug-drug interaction studies 
and bioequivalence studies, on the other hand, often utilize a fixed sequence design or a crossover design. In terms 
of hypothesis testing problems, the two most common ones are one-sided significance test and equivalence test 
versus two-sided significance tests. Although Pharmacodynamics (PD) endpoints are routinely collected in clinical 
pharmacology studies, most hypothesis testing problems are for PK endpoints and this is the main focus of this 
paper. The reader can easily adapt the methods presented here for power analysis of PD hypotheses.  

For hypotheses formulated through PK parameters, the parameters of interest are typically AUC, Cmax, or Ctrough. 
Since experience has shown that those PK parameters are lognormally distributed, the hypotheses are typically 
stated using the population geometric mean (or equivalently, the population median) of the pharmacokinetic 
parameters. For example, in a multiple dose study, the hypothesis may be stated as "At one or more well tolerated 
doses, the steady state population geometric mean Ctrough exceeds 35 nM"; in a bioequivalence study, the hypothesis 
may be stated as "The population geometric mean AUC0- ∞  ratio (Formulation 2 / Formulation 1) is between 0.80 and 
1.25". Note that, because of the lognormality assumption, PK hypotheses in clinical pharmacology studies can also be 
stated in terms of the population mean or mean difference of the log-transformed PK parameters. For the previous 
examples, "population mean Ctrough exceeds 35 nM" is equivalent to "Population mean ln(Ctrough) exceeds ln(35 nM)"; 
"population geometric mean AUC0- ∞  ratio is between 0.80 and 1.25" is equivalent to "difference in population mean 
ln(AUC0- ∞ ) is between ln(0.80) and ln(1.25)". Throughout this paper, the power analyses will be discussed in the log 
scale. As a result, the parameter of interest is either a population mean or a population mean difference. This makes 
inputting parameters for power calculations more straightforward, hence less error-prone. 

The Proc Power procedure was first introduced in SAS 9.1 to perform prospective power and sample size analysis for 
a variety of statistical methods, including t-test, equivalence test, one way analysis of variance, logistic regression with 
binary response, etc. In this paper, we will demonstrate that, for most clinical pharmacology studies, one can use the 
one- and two-sample t-tests and equivalence tests in Proc Power to perform model-based power calculations. 
Because using Proc Power involves reformulating a more complex problem into a simple statistical test, one needs a 
deeper understanding of the underlying statistical model. To that end, the relevant theoretical results and 
methodologies will be presented in Section 2. This is followed by several clinical pharmacology study examples to 
illustrate the implementation of different model-based power analysis in Section 3, illustration of power assessment for 
higher order crossover designs in Section 4, and further discussion in Section 5.  



 

2. THEORETICAL BACKGROUND 
We start by reviewing the noncentral t distribution. 

Under the statistical models used in most clinical pharmacology studies, the test statistic follows a t distribution 
(central or non-central, depending on the true parameter value). In order to simplify the discussion, we adopt the 
approach of Berger and Hsu (1996) and present the construction of the t statistic in the context that is independent of 
the underlying study design. In general, the construction of a t statistic involves two sample statistics. The first 

statistic T has a normal distribution with mean μ and variances .Tσ 2 The second statistic, denoted as SE( ),T  

estimates the standard error of .T  SE( )T is independent of T and 2 2 2[SE( )] / ( ),Tr T rσ χ∼  where the degree of 
freedom r is a positive number that depends on the study design. It follows that, for any real number ,a  

We next present several examples to illustrate how T and SE( )T can be constructed in some common study designs, 
including fixed sequence, parallel group, one-way analysis of variance (ANOVA), and 2 x 2 crossover designs. Those 
examples will serve as the theoretical foundation for power calculations in those settings. Power analysis for higher 
order crossover designs will be discussed separately in Section 4. 

2.1 EXAMPLES FOR CONSTRUCTION OF T STATISTICS 

Design 1: Two-period Fixed Sequence Design 

The data consists of bivariate measurements 1 2( , ),  1, , ,i iX X i n= … such that 2
2 1 ( , ).

IID

i i D DX X N μ σ− ∼ We therefore 

have 2
2 1 ( , / ),D DX X N nμ σ− ∼ where 1X and 2X are sample means for period 1 and period 2 data, respectively. 

Independently, 2 2 2( 1) / ( 1),D Dn S nσ χ− −∼ where 2
DS is the sample variance of 2 1{ ,  1, , }.i iX X i n− = …  Therefore, (1) 

Design 2: Two-sample Parallel Group Design with Common Variance 

Group 1 consists of 1n data points 2
1 1 1( , ),  1, , ;

IID

iX N i nμ σ =∼ … independently, group 2 consists of 2n data points 

where 2
pS is the pooled sample variance 

with 2
1S and 2

2S being the sample variance of group 1 and group 2, respectively. Therefore, (1) holds with 

 

2 1

2

1 2

2 1

2 2

1 2

1 2

,

1 1( ) ( ) ,

,

1 1( ) ,

2.

p

T

T X X

SE T S
n n

n n

r n n

μ μ μ

σ σ

≡ −

≡ +

≡ −

≡ +

≡ + −

 

Design 3: 2 X 2 Crossover Design 

In a 2 x 2 crossover study, two sequences of subjects are enrolled. Subjects in sequence 1 receive treatment A first, 
and after a suitable washout period, crossover to treatment B. Subjects in sequence 2, on the other hand, receive 
treatment B first, then crossover to A. Let ijky denote the measurement for subject j in sequence i collected in period 

 ( 1, ,  for 1,2, 1,2).ik j n i k= = =…  It is assumed that 

 ( , ) ,ijk i k d i k ij ijky g sμ π τ ε= + + + + +  

where 

• :ig  Effect associated with group (sequence) .i  

• :kπ Effect associated with period .k  

• :( , )d i kτ  Effect of treatment ( , )d i k - the treatment in sequence i of period k (e.g, (1,1)d A= ). 

• :ijs  Subject effect with 2(0, ).
IID

ij Ss N σ∼  

• :ijkε  Residual error with 2(0, ),
IID

ijk Nε σ∼ that is independent to .ijs  

Based on the above model, the populations mean of ijky by sequence and period is summarized in the table below: 

Sequence Period 1 Period 2 

1 (AB) 1 Agμ π τ1+ + +  
1 2 Bgμ π τ+ + +  

2 (BA) 
2 Bgμ π τ1+ + +  2 2 Agμ π τ+ + +  

 

Following Hills and Armitage (1979), the treatment comparison can be analyzed as a two-sample t-test. Specifically, if 

we let 1 1 2 1 1

2.2 FORMULA FOR POWER CALCULATIONS 
We next consider the derivation of the power functions in the general framework. We focus on three tests: the lower 
one-sided test, the upper one-sided test, and the equivalence test. 

For a lower one-sided test, under the alternative hypothesis, the true parameter value μ lies to the left of the null 

value 0 .μ Specifically, the hypothesis testing problem is stated as: 

For a size- α test, the power function can be easily derived: 

where  ,1rt α− is defined by ,1Pr( ( ) ) 1rt r t α α−≤ = − .  
In an upper one-sided test, under the alternative hypothesis, the true parameter value μ lies to the right of the null 

value 0μ and the hypothesis takes the form: 

0 0:  versus .H μ μ μ μ0≤ >  

For a size- α test, the power function is: 

In an equivalence test, under the alternative hypothesis, the true parameter value μ falls between two pre-specified 

values Lμ and .Uμ Hence the hypothesis takes the form: 

 0 1:  or  versus : .                 (2)L U L UH Hμ μ μ μ μ μ μ≤ ≥ < <  

Westlake (1972) and Schuirman (1987) proposed what has become the standard test of (2). It is called the "two one-
sided tests" (TOST). Following the TOST procedure, for a size- α test, one constructs the (1 2 )100%α− confidence 

interval for μ and rejects the null hypothesis if the confidence interval falls entirely between Lμ and .Uμ Therefore the 
power function can be expressed as 

Applying law of iterative expectations, Phillips (1990) derived the power function: 

is referred to as Owen's Q function. In the definition of  Owen's Q function, ( )φ i and ( )Φ i are the pdf and cdf of 
standard normal distribution, respectively. 

3. SOME EXAMPLES 

EXAMPLE 1: EQUIVALENCE TEST IN A TWO-PERIOD FIXED SEQUENCE STUDY 
Consider a midazolam interaction study to be conducted using a two-period fixed-sequence design. In period 1, 
midazolam will be administered as a single dose. After suitable washout, in period 2, the candidate drug C will be 
administered daily for seven days, and on Day 7, the same single dose midazolam will be co-administered with the 



 

candidate drug. The parameter of interest is midazolam AUC0- ∞  and the hypothesis is stated as "The true midazolam 
geometric mean AUC0- ∞  ratio (Period 2 / Period 1) lies within (0.5, 2.0)". 

We adopt the notation and model of Design 1 and let 1iX and 2iX represent midazolam ln(AUC0- ∞ ) in period 1 and 2, 
respectively. Then the hypothesis of interest can be formally stated as: 

 0 1: ln(2) or ln(0.5) versus : ln(0.5) ln(2).             D DH Hμ μ μ≤ ≥ < <  

Although the analysis is in nature a paired t test (or equivalently, a one-sample t test), ln(AUC0- ∞ ) will be formally 
analyzed by a linear mixed effect model with period as fixed effect and subject as random effect. Specifically, 

 ,ij j i ijX sμ ε= + +  

where 2(0, )
IID

i Ss N σ∼ and independently 2(0, ).
IID

ij Nε σ∼ This implies that 

2
2 1 1 2var( ) var( ) 2 .D i i i iX Xσ ε ε σ 2= − = − =  

Based on a previous study, the within-subject variance σ 2 is estimated to be 0.0735. This implies that Dσ can be 

estimated as 2 0.0735 0.3834.⋅ = The objective here is to provide power for n=6, 7, 8 under the assumption 0Dμ = (or 
equivalently, the true geometric mean ratio is equal to 1.0). This can be accomplished by regarding the problem as a 
one-sample equivalence test on 2 1i iX X−  and using the "onesamplemeans" option in proc power:  

%let sigma2=0.0735; * macro variable for the within-subject variance; 
%let sigma_D=%sysevalf((2*&sigma2)**0.5); * macro variable for the standard 
deviation of the difference; 
%let log_pt_5=%sysfunc(log(0.5)); * macro variable for log(0.5); 
%let log_2=%sysfunc(log(2)); * macro variable for log(2); 
proc power;  
   onesamplemeans test=equiv alpha=0.05 
   lower=&log_pt_5 upper=&log_2 std=&sigma_D 
   mean=0 
   ntotal=6 7 8 
   power =.;  
run; 
 

The SAS code is rather self-explanatory. The keyword "test=equiv" specifies that the power analysis is for an 
equivalence test. The equivalence bounds (ln(0.5) and ln(2)) are specified by the keywords "lower" and "upper", 
respectively. The standard deviation of 2 1i iX X− is communicated through the "std" keyword. Note that the SAS 
macro facility is extensively used to simplify inputting parameters for the power analysis. The result of the power 
calculation is displayed below. 

    Computed Power 
 
         N 
Index    Total Power 
 
1        6     0.965 
2        7     0.989 
3        8     0.996 

Hence the computed powers for N=6, 7, 8 are 0.965, 0.989, and 0.996, respectively. 

EXAMPLE 2: UPPER ONE-SIDED TEST IN FIXED-SEQUENCE STUDY 
Consider a rifampin interaction study to be conducted using a fixed-sequence design. In period 1, the candidate drug 
C will be administered as a single dose. After suitable washout, in period 2, single doses of 600 mg rifampin will be 
administered daily for 14 days, and on Day 14, 600 mg rifampin will be coadiminstered with the candidate drug. The 
parameter of interest is the AUC0- ∞ for the candidate drug and the hypothesis is stated as "The true geometric mean 
AUC0- ∞  ratio (Period 2 / Period 1) for the candidate drug C is greater than 0.5."  

We again adopt the notation and model of Design 1 and let 1iX and 2iX represent midazolam ln(AUC0- ∞ ) in period 1 
and 2, respectively. Then the hypothesis of interest can be formally stated as: 

 0 1: ln(0.5) versus : ln(0.5).              D DH Hμ μ≤ >  



 

The objective here is to estimate the sample size that gives 90% power for a true GMR of 0.65. From a previous 
study, the within-subject variance σ 2 is estimated to be 0.0408. Since in the hypothesis testing framework, this is an 
upper one-sided test, one can use the "onesamplemeans" option with the keyword "sides=U", which specifies an 
upper one-sided test, to perform the power calculation. 

%let sigma2=0.0408; * macro variable for the within-subject variance;   
%let sigma_D=%sysevalf((2*&sigma2)**0.5); * macro variable for the standard 
deviation of the difference; 
%let log_pt_5=%sysfunc(log(0.5)); * macro variable for log(0.5); 
%let log_true_gmr=%sysfunc(log(0.65)); * macro variable for log(0.65);  
proc power;  
   onesamplemeans test=t sides=U alpha=0.05 
   nullmean=&log_pt_5 std=&sigma_D 
   mean=&log_true_gmr 
   ntotal=. 
   power =0.9;  
run; 

The SAS output is displayed below: 

Computed N Total 
 
Actual    N 
 Power    Total 
0.909       12 

Hence a total of 12 subjects are needed to achieve an actual power of 90.9%. 

EXAMPLE 3: EQUIVALENCE TEST IN 2 X 2 CROSSOVER STUDY 
Consider a definitive bioequivalence study to be conducted using a 2 x 2 crossover design, with the two treatments 
being two formulations (denoted as A and B) of the candidate drug C. The parameter of interest is AUC0- ∞ and the 
hypothesis is stated as "The true geometric mean AUC0- ∞ ratio for the candidate drug C (Formulation B / Formulation 
A) is between 0.80 and 1.25." We adopt the notation and model of Design 3 and let ijky denote the ln(AUC0- ∞ ) value 

collected at period k for the jth subject in sequence i. The hypothesis can therefore be formally stated as: 

0 1: ln(0.80) or ln(1.25) versus : ln(0.80)< ln(1.25).B A B A B AH Hτ τ τ τ τ τ− ≤ − ≥ − <  

In this study, the sample sizes in the two sequences will be equal ( 1 2n n n= = ). It is expected that the true geometric 
mean ratio (GMR) is 1.1. The objective is therefore to choose a sample size so that there is at least 95% probability to 
conclude equivalence given the true GMR is 1.1. Based on Design 3, the power analysis can be conducted using a 

two-sample t test with n subjects per group and a common variance 
1

.
2

σ 2  From a prior study, the within-subject 

variance σ 2 is estimated to be 0.1003 and the "twosamplemeans" option can be used to perform the power analysis: 

%let sigma2=0.1003; * macro variable for the within-subject variance; 
%let std_derived=%sysevalf((&sigma2/2)**0.5); * macro variable for the derived 
common standard deviation; 
%let log_pt_8=%sysfunc(log(0.8)); * macro variable for log(0.8); 
%let log_1_pt_25=%sysfunc(log(1.25)); * macro variable for log(1.25); 
%let log_true_gmr=%sysfunc(log(1.1)); * macro variable for log(1.1); 
proc power;  
   twosamplemeans test=equiv_diff alpha=0.05 
   lower=&log_pt_8 upper=&log_1_pt_25 std=&std_derived 
   meandiff=&log_true_gmr 
   npergroup=. 
   power =0.95;  
run; 

Note that the keyword "test=equiv_diff" specifies an equivalence test for the population mean difference of two 
independent samples. The SAS output is displayed below. 

Computed N Per Group 
 
Actual    N Per 
 Power    Group 
 0.952       68 

Based on the calculation, a sample size of 68 per treatment sequence is needed to achieve an actual power of 
95.2%. 



 

EXAMPLE 4: LOWER ONE-SIDED TEST IN 2 X 2 CROSSOVER STUDY 
Consider a ketoconazole interaction study to be conducted using a 2 x 2 crossover design. Denote the two treatments 
as A and B, respectively. For treatment A, 300 mg of the candidate drug C will be administered as a single dose. In 
treatment B, single doses of 400 mg kenotconazole will be administered daily for 5 days and a single 300 mg dose of 
the drug C will be co-administered on Day 2. The parameter of interest is AUC0- ∞ for the candidate drug and the 
hypothesis is stated as "The true geometric mean AUC0- ∞ ratio for the candidate drug C (B / A) is less than 2.0. "  We 
adopt the notation and model of Design 3 and let ijky denote the ln(AUC0- ∞ ) value collected at period k for the jth 

subject in sequence i. The hypothesis can therefore be formally stated as: 

0 1: ln(2.0)  versus : ln(2.0).B A B AH Hτ τ τ τ− ≥ − <  

The objective is to identify the true GMR that gives 80% power for n=4 per sequence. The within-subject variance 
σ 2 is estimated to be 0.03821. We therefore use the "twosamplemeans" option with the keyword "sides=L", which 
specifies a lower one-sided test, to perform the power calculation. 

%let sigma2=0.03821; * macro variable for the within-subject variance; 
%let std_derived=%sysevalf((&sigma2/2)**0.5); * macro variable for the derived 
common standard deviation;  
%let log_2=%sysfunc(log(2.0)); * marco variable for log(2.0); 
%let log_true_gmr=%sysfunc(log(1.518)); * macro variable for log(1.518); 
proc power;  
   twosamplemeans test=diff side=L alpha=0.05 
   nulldiff=&log_2 std=&std_derived 
   meandiff=&log_true_gmr 
   npergroup=4 
   power =.;  
run; 

Note that Proc Power does not allow direct calculation of true mean difference given power and sample size. Hence in 
this example, the true GMR that gives 80% power can only be obtained by trial and error. The SAS output is 
displayed below. 

Computed Power 
 
    Power 
    0.800 

Therefore for n=4 per sequence and a true GMR of 1.518, the power is 80%. 

4. POWER CALCULATION FOR HIGHER ORDER CROSSOVERS 
First consider a 3 period crossover design with 3 treatments (denoted as A, B, and C) and 6 treatment sequences 
(see Diagram (a)). We adopt the same model (i.e, treatment and period as fixed effects, and subject as a random 
effect) and notation as those in Design 3. Without loss of generality, assume the parameter of interest is .B Aτ τ− Let 

T denote the estimator for .B Aτ τ− Using linear model theory, it can be shown that under the assumption of equal 
number of subjects per sequence, T can be expressed: 

 
( , ) ( , )

1 1
,

6 6ijk ijk
d i k B d i k A

T y y
n n= =

where n denote the number of subjects per treatment sequence. It follows that 2

3T n
σσ

2

= , where σ 2 is the within subject 

variance. 

Diagram (a). Three Period Crossover Design 

Sequence   Period   
  1 2 3 
1 A B C 
2 A C B 
3 B A C 
4 B C A 
5 C A B 
6 C B A 

 



 

The error degree of freedom r for the 3 period crossover design is: 

(18 1) 2 2 (6 1)

12 4.

total period trt sujbectr df df df df

n n

n

= − − −

= − − − − −

= −

 

All power calculations can be conducted through the "onesamplemeans" option in Proc Power by formulating the 
problem as a one-sample t test or equivalence test. The idea is to modify the input for "ntotal" and "std" so that the 
correct degrees of freedom r and variance 2

Tσ are entered. In a one-sample t test or equivalence test, the degrees of 
freedom is calculated as ntotal – 1. This suggests that we should set "ntotal" to 12 3,n − which ensures that the 
degrees of freedom is equal to 12 4.n −  Since for a one-sample t test or equivalence test with "ntotal" set to 12 3,n −  

the variance of the estimator is calculated as 
2std

,
12 -3n

the correct input for "std" can be obtained by solving the 

following equation: 

 
2

2std
.

12 -3 3Tn n
σ

σ
2

= =  

Hence "std" should be set to 
1/ 24 1n

n
σ−⎛ ⎞ .⎜ ⎟

⎝ ⎠
 

For crossover designs with more than 3 periods, to be specific, we focus on the class of Williams designs (Williams 
(1949)), which are widely used in practice.  Examples of layouts for 4, 5, and 6 period Williams designs are displayed 
in Diagrams (b), (c), and (d) respectively. 

Diagram (b). Four Period Williams Design 

Sequence   Period     
  1 2 3 4 
1 A D B C 
2 B A C D 
3 C B D A 
4 D C A B 

 

 

Diagram (c). Five Period Williams Design 

Sequence   Period       
  1 2 3 4 5 
1 A E B D C 
2 B A C E D 
3 C B D A E 
4 D C E B A 
5 E D A C B 
6 A B E C D 
7 B C A D E 
8 C D B E A 
9 D E C A B 

10 E A D B C 
 



 

Diagram (d). Six Period Williams Design 

Sequence   Period         
  1 2 3 4 5 6 
1 A  F B E C D 
2 B A C F D E 
3 C B D A E F 
4 D C E B F A 
5 E D F C A B 
6 F E A D B C 

 

For the designs in (b), (c), and (d), provided that there are equal number of subjects in each sequence, the estimator 
of B Aτ τ− can be expressed in the form of: 

( , ) ( , )

1 1
,ijk ijk

d i k B d i k A
T y y

sn sn= =
∑ ∑= −  

where n denote the number of subjects per treatment sequence and s the number of sequences. Hence the 
"onesamplemeans" option in Proc Power can again be used to perform power calculations and the appropriate input 
for "ntotal" and "std" can be similarly derived. The results are summarized in the table below. 

Design ntotal std 

Three period (six sequences)  12 3n−  
 

1/24 1n
n

σ−⎛ ⎞
⎜ ⎟

Four period Williams design  12 5n −  
 

1/ 212 5
2
n
n

Five period Williams design  40 7n −  

 
1/ 240 7

5
n
n

Six period Williams design  30 9n −  
 

1/ 230 9

n
n

We next present two examples. 

EXAMPEL 5: EQUIVALENCE TEST IN 3 PERIOD CROSSOVERS 
Consider a formulation study for the candidate drug C to be conducted using the 3 period crossover design in 
Diagram (a).  The three treatments are Phase I formulation, Phase II formulation, and Phase III formulation.  The 
parameter of interest is AUC0- ∞ for the candidate drug and the primary hypothesis is stated as "The true geometric 
mean AUC0- ∞ ratio for the candidate drug C (Phase III formulation / Phase II formulation) is between 0.70 and 1.43. " 
The study plans to enroll equal number of subjects per treatment sequence. Let ijky denote the ln(AUC0- ∞ ) value 

collected at period k for the jth subject in sequence i. The objective is to calculate the power given n=3 per treatment 
sequence and a true GMR of 1.20. The within-subject variance σ 2 is estimated to be 0.0389. By the discussion in 
Design 4, the power analysis can be conducted using the "onesamplemeans" option by setting "ntotal" to 12 3n−  and 

"std" to 
1/24 1 ,n

n
σ−⎛ ⎞

⎜ ⎟
⎝ ⎠

where n is the number of subject per treatment sequence. 

%let n_per_seq=3; * macro variable the number of subjects per treatment sequence; 
%let sigma2=0.0389; * macro variable for the within subject variance; 
%let ntotal_derived=%sysevalf(12*&n_per_seq-3); * macro variable for the derived 
ntotal in a 3 period crossover; 
%let std_derived=%sysevalf(((4-1/&n_per_seq)*&sigma2)**0.5); * macro variable for 
the derived std in a 3 period crossover; 
%let log_pt_7=%sysfunc(log(0.70)); * macro variable for log(0.70); 
%let log_1_pt_43=%sysfunc(log(1.43)); * macro variable for log(1.43); 
%let log_true_gmr=%sysfunc(log(1.20)); * marco variable for log(1.20); 
proc power; 
 onesamplemeans test=equiv alpha=0.05 
 lower=&log_pt_7 upper=&log_1_pt_43 std=&std_derived 
 mean=&log_true_gmr 



 

 ntotal=&ntotal_derived 
 power =.; 
run; 

 
The power is 83.3% for n=3 per sequence and a true GMR of 1.20. 

Computed Power 
 
    Power 
 
    0.833 

EXAMPEL 6: UPPER ONE-SIDED TEST IN 5 PERIOD CROSSOVERS 
Consider a formulation study for the candidate drug C that compares one old formulation with four new formulations. 
The study will be conducted using a 5 period crossover design (Diagram (c)) and the parameter of interest is AUC0-

∞ for the candidate drug. The primary hypothesis can be stated as "The true geometric mean  AUC0- ∞ ratio (New 
formulation 1 / Old formulation) is above 0.70". The study plans to enroll equal number of subjects per treatment 
sequence. Let ijky denote the ln(AUC0- ∞ ) value collected at period k for the jth subject in sequence i. The objective is 

to calculate the power given n=2 per treatment sequence and a true GMR of 0.80. The within-subject variance σ 2 is 
estimated to be 0.0285. By the discussion in Design 4, the power analysis can be conducted using the 

"onesamplemeans" option by setting "ntotal" to 40 7n −  and "std" to 
1/ 240 7 ,

5
n
n

σ−⎛ ⎞
⎜ ⎟
⎝ ⎠

where n is the number of subject 

per treatment sequence. 

%let n_per_seq=2; * macro variable the number of subjects per treatment sequence; 
%let sigma2=0.0285; * macro variable for the within subject variance; 
%let ntotal_derived=%sysevalf(40*&n_per_seq-7); * macro variable for the derived 
ntotal in a 5 period crossover; 
%let std_derived=%sysevalf(((8-7/5/&n_per_seq)*&sigma2)**0.5); * macro variable for 
the derived std in a 5 period crossover; 
%let log_pt_7=%sysfunc(log(0.70)); * marco variable for log(0.70); 
%let log_true_gmr=%sysfunc(log(0.80)); * macro variable for log(0.80); 
proc power; 
 onesamplemeans test=t side=U alpha=0.05 
 nullmean=&log_pt_7 std=&std_derived 
 mean=&log_true_gmr 
 ntotal=&ntotal_derived 
 power =.; 
run; 

The power is 79.8% for n=2 per sequence and a true GMR of 0.80. 

Computed Power 
 
    Power 
    0.798 

5. DISCUSSION 
In this paper, we demonstrate that many hypothesis testing problems in clinical pharmacology studies can be 
reformulated as a one- or two-sample t-test or equivalence test. Thereby one can use SAS Proc Power to perform the 
power analysis.  Sample programs for different hypothesis testing problems are presented in a variety of study design 
settings.  

Besides SAS, NQuery and PASS are also widely used in practice for power calculations. The power analyses for one 
and two-sample t-tests and equivalence tests in two-sample parallel group design and 2 x 2 crossover design are 
straightforward in both packages. However, neither packages offer a direct solution for power calculations in one-
sample equivalence tests1. To the extent possible, the calculations in all examples have been validated using NQuery 
and PASS. We conclude this paper by outlining several advantages of using SAS Proc Power versus NQuery and 
PASS to perform power analysis. First, since one often obtains variance estimates by analyzing relevant clinical data 
in SAS, it is more streamlined and convenient to conduct power analysis in SAS as well. Second, the documentation 
in SAS is more comprehensive than those in NQuery and PASS: this helps the users better understand the procedure 
and hence the users are less likely to make mistakes. Third, as demonstrated in the examples, one can take 
advantage of the programming capabilities in SAS and make inputting parameters for power analysis more 

                                                           
1 NQuery has no such capability directly.  In order to cover more distributions, PASS uses a simulation approach even though a 
closed-form solution is available for one-sample equivalence test based on t statistic. 



 

straightforward. Last, but not least, using SAS can better help the users keep a record of the power analysis, which is 
helpful for adapting the code for similar problems in the future. 

